<?xml version="1.0" encoding="UTF-8" ?>
<xsl:stylesheet xmlns:xsl="http://www.w3.org/1999/XSL/Transform" version="1.0" name="Основной шаблон">

  <xsl:output doctype-public="-//W3C//DTD HTML 4.01//EN" doctype-system="http://www.w3.org/TR/html4/strict.dtd" />


  <!-- Каркас страницы -->
  <xsl:template match="/node()">
    <html lang="{@lang}">
      <head>
        <title><xsl:value-of select="head/title" /></title>
        <meta name="keywords" lang="{@lang}" content="{head/keywords}"/>
        <meta name="description" lang="{@lang}" content="{head/description}" />
        <link rel="Shortcut Icon" href="/favicon.ico"/>
        <!-- имя таблицы стилей должно совпадать с именем шаблона -->
        <link rel="stylesheet" type="text/css" href="/admin/templates/main.css" />
      </head>
      <body>

        <xsl:call-template name="body" />
      </body>
    </html>
  </xsl:template>


  <!-- внутренности -->
  <xsl:template name="body">
    <div align="center"><div class="main">

    <div class="head">
      <h2><a href='/' title='Smarta - управление цветом'>Smarta</a> - управление цветом</h2>
    </div>

    <xsl:apply-templates select="nav"/>

    <hr class="oldbrowsers"/>

    <div class="left">
      <p><xsl:apply-templates select="breadcrumb"/></p>
      <h3>Содержимое</h3>
      <xsl:apply-templates select="content"/>
      <xsl:apply-templates select="error"/>
    </div>

    <hr class="oldbrowsers"/>

    <div class="right">
      <h3>Подразделы</h3>
      <xsl:apply-templates select="offsprings"/>
    </div>

    <hr class="oldbrowsers"/>
    <div class="clear">&#160;</div>
    <div class="copyright">
    	<xsl:call-template name="time"/>
    </div>

    </div></div>

  </xsl:template>


  <!-- content -->
  <xsl:template match="content">
    <p><xsl:value-of select="." disable-output-escaping="yes" /></p>
  </xsl:template>


  <!-- хлебные крошки -->
  <xsl:template match="breadcrumb">
    <div class="breadcrumb"><xsl:apply-templates select="item" /></div>
  </xsl:template>

  <!-- одна крошка -->
  <xsl:template match="breadcrumb/item">
    <xsl:if test="@current='true'">
      <span><xsl:value-of select="." /></span>
    </xsl:if>
    <xsl:if test="not (@current='true')">
      <a href="{@href}" title="{@title}"><xsl:value-of select="." /></a>
    </xsl:if>
    <xsl:if test="not (position()=last())">
      <xsl:text> » </xsl:text>
    </xsl:if>
  </xsl:template>


  <!-- основная навигация -->
  <xsl:template match="nav">
    <div class="nav"><xsl:apply-templates select="item" /></div>
  </xsl:template>

  <!-- навигационная ссылка -->
  <xsl:template match="nav/item">
    <xsl:if test="@current='true'">
      <span><xsl:value-of select="." /></span>
    </xsl:if>
    <xsl:if test="not (@current='true')">
      <a href="{@href}" title="{@title}"><xsl:value-of select="." /></a>
    </xsl:if>
    <xsl:if test="not (position()=last())">
      <xsl:text> | </xsl:text>
    </xsl:if>
  </xsl:template>


  <!-- подразделы -->
  <xsl:template match="offsprings">
    <ul><xsl:apply-templates select="item" /></ul>
  </xsl:template>

  <!-- ссылка подразделов -->
  <xsl:template match="offsprings/item">
    <li><a href="{@href}" title="{@title}"><xsl:value-of select="." /></a></li>
  </xsl:template>


  <!-- сообщене об ошибке -->
  <xsl:template match="error">
    <h3>Ошибка: <xsl:value-of select="." /></h3>
  </xsl:template>

  <!-- статистика работы движка -->
  <xsl:template name="time">
    <small>time2time2 timetime</small>
  </xsl:template>

</xsl:stylesheet>
